Gologit2: A Program for Generalized Logistic Regression/ Partial Proportional Odds Models for Ordinal Dependent Variables Richard Williams, <u>Richard.A.Williams.5@ND.Edu</u> Last revised May 12, 2005

[This document is a work in progress. Comments are welcome. Parts of this paper are adapted from the documentation for Vincent Fu's original gologit command and are used with permission. Those who learn best by examples may wish to skim over the early sections.]

Overview. gologit2 is a user-written program that estimates generalized logistic regression models for ordinal dependent variables. The actual values taken on by the dependent variable are irrelevant except that larger values are assumed to correspond to "higher" outcomes.

A major strength of gologit2 is that it can also estimate two special cases of the generalized model: the *proportional odds model* and the *partial proportional odds model*. Hence, gologit2 can estimate models that are less restrictive than the proportional odds /parallel lines models estimated by ologit (whose assumptions are often violated) but more parsimonious and interpretable than those estimated by a non-ordinal method, such as multinomial logistic regression (i.e. mlogit). The autofit option greatly simplifies the process of identifying partial proportional odds models that fit the data, while the pl (parallel lines) and npl (non-parallel lines) options can be used when users wish to specify the model themselves.

An alternative but equivalent parameterization of the model that has appeared in the literature is reported when the gamma option is selected. Other key advantages of gologit2 include support for linear constraints, Stata 8.2 survey data (svy) estimation, and the computation of estimated probabilities via the predict command.

gologit2 is inspired by Vincent Fu's gologit program and is backward compatible with it but offers several additional powerful options. gologit2 was written for Stata 8.2 and many of the references in the help file are for Stata 8 manuals.

Description. The ologit command included with Stata imposes what is called the *proportional odds assumption* on the data. This is also known as the *parallel lines/ parallel regressions assumption*. The proportional odds/parallel lines model is a special case of the generalized model estimated by gologit2. By default, gologit2 relaxes the proportional odds assumption and allows the effects of the explanatory variables to vary with the point at which the categories of the dependent variable are dichotomized. However, if the pl option is specified without parameters, gologit2 estimates the proportional odds model, e.g. the commands

```
ologit y x1 x2 x3
```

and

gologit2 y x1 x2 x3, pl lrforce

will produce equivalent results.

In practice, the proportional odds assumption is often violated by the data. Standard advice in such situations is to go to a non-ordinal model, such as mlogit. Unfortunately, such models can be far less parsimonious and more difficult to interpret than the proportional odds model. gologit2 provides an alternative by estimating partial proportional odds models. With such models, the parallel lines/ proportional odds assumption can be relaxed for some explanatory variables while being maintained for others. For example, the command

gologit2 y x1 x2 x3, npl(x1)

would relax the proportional odds/parallel lines assumption for x1 while maintaining it for x2 and x3. An equivalent command is

gologit2 y x1 x2 x3, pl(x2 x3)

which forces x^2 and x^3 to meet the proportional odds/ parallel lines assumption while not imposing the assumption on x^1 .

More formally, suppose we have an ordinal dependent variable Y which takes on the values 1, 2, ..., m. The generalized ordered logit model estimates a set of coefficients (including one for the constant) for each of the m - 1 points at which the dependent variable can be dichotomized. The probabilities that Y will take on each of the values 1, ..., m is equal to

 $\begin{array}{ll} P(\;Y=1\;)=F(\;-XB_1\;)\\ P(\;Y=j\;)=F(\;-XB_j\;)-F(\;-XB_{j-1}\;) & j=2,\,...,\,m-1\\ P(\;Y=m\;)=1\,-F(\;-XB_{m-1}\;) \end{array}$

The generalized ordered logit model uses the logistic distribution as the cumulative distribution, although other distributions may also be used. The logistic distribution allows researchers to interpret this model in terms of logits:

The proportional odds model (estimated by Stata's ologit command and by gologit2 with the pl option) restricts the B_k coefficients to be the same for every dividing point k = 1, ..., m-1. The partial proportional odds model (estimated in gologit2 via the npl() and pl() options) restricts some B_k coefficients to be the same for every dividing point while others are free to vary.

Note that unlike models such as OLS regression and binary logit, the generalized ordered logit model imposes explicit restrictions on the range of the X variables. Since probabilities are by definition constrained to be in the range [0,1], valid combinations of the X variables must satisfy the following inequalities:

$$XB_1 >= XB_2 >= XB_3 \dots >= XB_{m-1}$$

Other scholars (e.g. Peterson & Harrell, 1990) have proposed an alternative but equivalent parameterization of the partial proportional odds model in which there is only one set of Betas but a second set of coefficients, called Gammas, can vary across the dividing points. The gammas indicate the extent to which the proportional odds assumption does not hold for a variable; if the gammas for a variable equal 0, then the parallel lines assumption holds for that variable. This parameterization can be displayed by using the gamma option.

Key Options. gologit2 supports many standard Stata options, which work the same way as they do with other Stata commands. Options which are unique to gologit2 are described below. See the help file for other options. The complete syntax is

```
gologit2 depvar [indepvars] [weight] [if exp] [in range] [, lrforce pl pl(varlist) npl
npl(varlist) autofit autofit(alpha) gamma nolabel store(name)
constraints(clist) robust cluster(varname) level(#) score(newvarlist/stub*) or
log v1 svy svy_options maximize_options ]
```

pl, npl, npl(), pl(), autofit and autofit() provide alternative means for imposing or relaxing the proportional odds/ parallel lines assumption. Only one may be specified at a time.

- autofit(alpha) uses an iterative process to identify the partial proportional odds model that best fits the data. alpha is the desired significance level for the tests; alpha must be greater than 0 and less than 1. If autofit is specified without parameters, the default alpha-value is .05. Note that, the higher alpha is, the easier it is to reject the parallel lines assumption, and the less parsimonious the model will tend to be. This option can take a little while because several models may need to be estimated. The use of autofit is highly recommended but other options provide more control over the final model if the user wants it.
- pl specified without parameters constrains all independent variables to meet the proportional odds assumption. It will produce results that are equivalent to ologit.
- npl specified without parameters relaxes the proportional odds/ parallel lines assumption for all explanatory variables. This is the default option and presents results equivalent to the original gologit.
- pl(varlist) constrains the specified explanatory variables to meet the proportional odds/ parallel lines assumption. All other variable effects do not need to meet the assumption. The variables specified must be a subset of the explanatory variables.
- npl(varlist) frees the specified explanatory variables from meeting the proportional odds/ parallel lines assumption. All other explanatory variables are constrained to meet the assumption. The variables specified must be a subset of the explanatory variables.

lrforce forces Stata to report a Likelihood Ratio Statistic under certain conditions when it ordinarily would not. Some types of constraints can make a Likelihood Ratio chi-square test invalid. Hence, to be safe, Stata reports a Wald statistic whenever constraints are used. But, Likelihood Ratio statistics should be correct for the types of constraints imposed by the pl and npl commands. Note that the lrforce option will be ignored when robust standard errors are specified either directly or indirectly, e.g. via use of the robust or svy options. Use this option with caution if you specify other constraints since these may make a LR chi- square statistic inappropriate.

gamma displays an alternative but equivalent parameterization of the partial proportional odds model used by Peterson and Harrell (1990) and Lall et al (2002). Under this parameterization, there is one Beta coefficient and M-2 Gamma coefficients for each explanatory variable, where M = the number of categories for Y. The gammas indicate the extent to which the proportional odds assumption is violated by the variable, i.e. when the gammas do not significantly differ from 0 the proportional odds assumption is met. Advantages of this parameterization include the fact that it is more parsimonious than the default layout. In addition, by examining the test statistics for the Gammas, you can get a feel for which variables meet the proportionality assumption and which do not.

store(*name*) causes the command estimates store *name* to be executed when gologit2 finishes. This is useful for when you wish to estimate a series of models and want to save the results.

nolabel causes the equations to be named eq1, eq2, etc. The default is to use the first 32 characters of the value labels and/or the values of Y as the equation labels. Note that some characters cannot be used in equation names, e.g. the period (.), the dollar sign (\$), and the colon(:), and will be replaced with the underscore (_) character. The default behavior works well when the value labels are short and descriptive. It may not work well when value labels are very long and/or include characters that have to be changed to underscores. If the printout looks unattractive and/or you are getting strange errors, try changing the value labels of Y or else use the nolabel option.

vl causes gologit2 to return results in a format that is consistent with gologit 1.0. This may be useful/necessary for post-estimation commands that were written specifically for gologit (in particular, some versions of the Long and Freese spost commands support gologit but not gologit2). However, post-estimation commands written for gologit2 (including predict) may not work correctly if vl is specified.

log displays the iteration log. By default it is suppressed.

or reports the estimated coefficients transformed to relative odds ratios, i.e., exp(b) rather than b; see [R] ologit for a description of this concept. Options rrr, eform, and irr produce identical results (labeled differently) and can also be used.

constraints (*clist*) specifies linear constraints to be applied during estimation. Constraints are defined with the constraint command. constraints(1) specifies that the model is to be constrained according to constraint 1; constraints(1-4) specifies constraints 1 through 4; constraints(1-4,8) specifies 1 through 4 and 8. Keep in mind that the pl, npl and autofit options work by generating across-equation constraints, which may affect how any additional constraints should be specified. When using the constraint command, refer to equations by their equation #, e.g. #1, #2, etc.

svy indicates that gologit2 is to pick up the svy settings set by svyset and use the robust variance estimator. Thus, this option requires the data to be svyset; see help svyset. When using svy estimation, use of if or in restrictions will not produce correct variance estimates for subpopulations in many cases. To compute estimates for subpopulations, use the subpop() option. If svy has not been specified, use of other Stata 8.2 svy-related options (e.g. subpop, deff, meff) will produce an error.

Other standard Stata options supported by gologit2: robust cluster level score

Other standard svy-related options supported by gologit2: subpop nosvyadjust prob ci deff deft meff meft

Options available when replaying results: gamma store or level prob ci deff deft

prob, ci, deff and deft are only available when svy estimation has been used.

Options available for the predict command: xb stdp stddp p

p gives the predicted probability. Note that you specify one new variable with xb, stdp, and stddp and specify either one or k new variables with p. These statistics are available both in and out of sample; type "predict ... if e(sample) ..." if wanted only for the estimation sample.

Examples.

Example 1: Attitudes Toward Working Mothers. Long and Freese (2003) present data from the 1977/1989 General Social Survey. Respondents are asked to evaluate the following statement: "A working mother can establish just as warm and secure a relationship with her child as a mother who does not work." Responses were coded as 1 =Strongly Disagree (SD), 2 =Disagree (D), 3 =Agree (A), and 4 =Strongly Agree (SA). Explanatory variables are yr89 (survey year; 0 = 1977, 1 = 1989), male (0 =female, 1 =male), white (0 =nonwhite, 1 =white), age (measured in years) ed (years of education) and prst (occupational prestige scale). Based on their analysis (reproduced below), Long and Freese conclude that the parallel lines assumption is violated with these data and suggest that an alternative ordinal regression model or a multinomial

logit model may be called for (the brant command requires that the spost routines be installed):

. brant

Brant Test of Parallel Regression Assumption

Variable	chi2	p>chi2	df
All	49.18	0.000	12
yr89 male white age ed prst	13.01 22.24 1.27 7.38 4.31 4.33	0.001 0.000 0.531 0.025 0.116 0.115	2 2 2 2 2 2 2

A significant test statistic provides evidence that the parallel regression assumption has been violated.

The Brant test suggests that yr89 and male are especially problematic with regards to the parallel lines assumption, age is borderline, but the other variables do not appear to violate the assumption.

Using gologit2, we can (a) reproduce ologit's estimates by using the pl parameter, i.e. estimate a model in which all variables are constrained to meet the proportional odds/ parallel regressions/ parallel lines assumption, (b) estimate a model (gologit2's default) in which no

variables have to meet the parallel lines assumption (c) do a global likelihood ratio chi-square test of the parallel lines assumption, and (d) use autofit to estimate a model in which some variables are constrained to meet the parallel lines assumption while others are not.

Generalized O		LR ch	r of obs = i2(6) = > chi2 =	2293 301.72 0.0000				
Log likelihoo	d = -2844.9123	3		PIOD Pseud		0.0504		
<pre>(1) [SD]yr (2) [SD]ma (3) [SD]wh (4) [SD]ag (5) [SD]ed (6) [SD]pr (7) [D]yr8 (8) [D]mal (9) [D]whi (10) [D]age (11) [D]ed (12) [D]prs</pre>								
warm	Coef.	Std. Err.	 Z	P> z	[95% Conf.	Interval]		
SD yr89 male white age ed prst _cons	+	.0798989 .0784827 .1183808 .0024683 .015975 .0032929 .2389128	6.56 -9.34 -3.30 -8.78 4.20 1.84 10.32	0.000 0.000 0.001 0.000 0.000 0.065 0.000	.3673036 887123 6231816 0265032 .0358624 0003813 1.997102	.6805014 5794765 1591373 0168278 .0984831 .0125267 2.933622		
D yr89 male white age ed prst _cons	7332998 3911595 0216655	.0798989 .0784827 .1183808 .0024683 .015975 .0032929 .2333156	6.56 -9.34 -3.30 -8.78 4.20 1.84 2.70	0.000 0.000 0.001 0.000 0.000 0.065 0.007	.3673036 887123 6231816 0265032 .0358624 0003813 .1736138	.6805014 5794765 1591373 0168278 .0984831 .0125267 1.088194		
A yr89 male white age ed prst _cons	.5239025 7332998 3911595 0216655 .0671728 .0060727 -1.261854	.0798989 .0784827 .1183808 .0024683 .015975 .0032929 .234018	6.56 -9.34 -3.30 -8.78 4.20 1.84 -5.39	0.000 0.000 0.001 0.000 0.000 0.065 0.000	.3673036 887123 6231816 0265032 .0358624 0003813 -1.720521	.6805014 5794765 1591373 0168278 .0984831 .0125267 8031871		

Notice that, by imposing the parallel lines assumption, the same parameter estimates appear multiple times, because the effects are constrained to be equal for each cut-point (the constraints applied are explicitly stated in the top part of the printout). Also, the cut-points in ologit are constants (with signs reversed) in gologit2. The LR chi2 statistics are the same as in the

ologit model (the lrforce parameter told gologit2 to report a lr chi-square rather than the Wald chi-square that Stata would report by default). In short, even though the gologit2 and ologit output looks a little different, when the gologit2 pl parameter is used the exact same model is estimated by both. The use of the store option caused the results to be saved under the name "constrained" so we can use them for future hypothesis testing.

Now, we'll estimate a model in which no variables have to meet the parallel lines assumption (the npl parameter is explicitly specified here but it would have been used by default otherwise):

* Part b. No variables constrained to meet the pl assumption.

	. gologit2 warm yr89 male white age ed prst, npl lrforce store(unconstrained)							
	ered Logit 1 = -2820.31			LR ch Prob	er of obs = hi2(18) = > chi2 = lo R2 =	350.92		
wa	1rm	Coef.	Std. Err.	Z	P> z	[95% Conf	. Interval]	
ma whi a	ige ed ist	.95575 3009776 5287268 0163486 .1032469 0016912 1.856951	.1547185 .1287712 .2278446 .0039508 .0247377 .0055997 .3872576	6.18 -2.34 -2.32 -4.14 4.17 -0.30 4.80	0.000 0.019 0.020 0.000 0.000 0.763 0.000	.6525074 5533645 9752941 0240921 .0547619 0126665 1.09794	0485906 0821595 0086051 .151732	
ma whi a	ige ed ist	.5363707 717995 349234 0249764 .0558691 .0098476 .7198119	.0919074 .0894852 .1391882 .0028053 .0183654 .0038216 .265235	5.84 -8.02 -2.51 -8.90 3.04 2.58 2.71	0.000 0.012 0.000 0.002 0.002 0.010 0.007	.3562355 8933827 6220379 0304747 .0198737 .0023575 .1999609	.716506 5426072 07643 0194782 .0918646 .0173377 1.239663	
ma whi a	ge ed st	.3312184 -1.085618 3775375 0186902 .0566852 .0049225 -1.002225	.1127882 .1217755 .1568429 .0037291 .0251836 .0048543 .3446354	2.94 -8.91 -2.41 -5.01 2.25 1.01 -2.91	0.003 0.000 0.016 0.000 0.024 0.311 0.004	.1101577 -1.324294 684944 025999 .0073263 0045918 -1.677698	.5522792 8469423 070131 0113814 .1060441 .0144368 3267523	

We see that the estimates for yr89 and male (the variables which the Brant test said were most problematic) differ substantially across equations, while differences in the effects of other variables are fairly small. We can now do a global test of the proportional odds assumption by contrasting the two models we have just estimated:

```
. * Part c. Do a global test of the parallel lines assumption
. lrtest constrained unconstrained
likelihood-ratio test LR chi2(12) = 49.20
(Assumption: constrained nested in unconstrained) Prob > chi2 = 0.0000
```

The chi-square statistic (which is similar to the Brant statistic reported earlier, but should be more accurate because it is a likelihood ratio test rather than Wald) shows that *at least one variable* does not meet the parallel lines assumption. But, *it need not mean that all fail to meet the assumption.* Hence, we can now use the autofit option to see whether a partial proportional odds model can fit the data. In a partial proportional odds model, some variables meet the proportional odds assumption while others do not.

```
. * Part d. Use autofit to identify/estimate a partial proportional odds model that
fits the data
. gologit2 warm yr89 male white age ed prst, autofit lrf
   _____
Testing parallel lines assumption using the .05 level of significance...
Step 1: white meets the pl assumption (P Value = 0.7136)
Step 2: ed meets the pl assumption (P Value = 0.1589)
Step 3: prst meets the pl assumption (P Value = 0.2046)
Step 4: age meets the pl assumption (P Value = 0.0743)
Step 5: The following variables do not meet the pl assumption:
       yr89 (P Value = 0.00093)
       male (P Value = 0.00002)
If you re-estimate this exact same model with gologit2, instead
of autofit you can save time by using the parameter
pl(white ed prst age)
 _____
                                         Number of obs =
Generalized Ordered Logit Estimates
                                        Number of obs=2293LR chi2(10)=338.30Prob > chi2=0.0000Pseudo R2=0.0565
                                                            2293
Log likelihood = -2826.6182
(1) [SD]white - [D]white = 0
(2) [SD]ed - [D]ed = 0
(3) [SD]prst - [D]prst = 0
( 4) [SD]age - [D]age = 0
(5) [D]white - [A]white = 0
(6) [D]ed - [A]ed = 0
(7) [D]prst - [A]prst = 0
(8) [D]age - [A]age = 0
     _____
     warm | Coef. Std. Err. z P>|z| [95% Conf. Interval]
_____
SD
      yr89.98368.15300916.430.000.68378761.283572male-.3328209.1275129-2.610.009-.5827417-.0829002
     white -.3832583 .1184635 -3.24 0.001 -.6154424 -.1510742
      age -.0216325 .0024751 -8.74 0.000 -.0264835 -.0167814
       ed | .0670703 .0161311 4.16 0.000 .0354539 .0986866
     prst 0.0059146 .0033158 1.78 0.074 -.0005843 .0124135
_cons 2.12173 .2467146 8.60 0.000 1.638178 2.605282
______
```

D							
	yr89	.534369	.0913937	5.85	0.000	.3552406	.7134974
	male	6932772	.0885898	-7.83	0.000	8669099	5196444
	white	3832583	.1184635	-3.24	0.001	6154424	1510742
	age	0216325	.0024751	-8.74	0.000	0264835	0167814
	ed	.0670703	.0161311	4.16	0.000	.0354539	.0986866
	prst	.0059146	.0033158	1.78	0.074	0005843	.0124135
	_cons	.6021625	.2358361	2.55	0.011	.1399323	1.064393
	4	+					
A							
	yr89	.3258098	.1125481	2.89	0.004	.1052197	.5464
	male	-1.097615	.1214597	-9.04	0.000	-1.335671	8595579
	white	3832583	.1184635	-3.24	0.001	6154424	1510742
	age	0216325	.0024751	-8.74	0.000	0264835	0167814
	ed	.0670703	.0161311	4.16	0.000	.0354539	.0986866
	prst	.0059146	.0033158	1.78	0.074	0005843	.0124135
	_cons	-1.048137	.2393568	-4.38	0.000	-1.517268	5790061

The results show that 4 of the 6 variables (white, age, ed, prst) meet the parallel lines assumption. Only yr89 and male do not. This model is less restrictive than the model estimated by ologit (whose assumptions are violated in this case) but much more parsimonious than a non-ordinal alternative such as mlogit.

Interpetation. Of course, now that you've got these parameters, how do you interpret them? In general, you can interpret gologit2 coefficients as coefficients from binary logit models where you have collapsed the categories of your outcome variable into two categories. Suppose your categories are numbered 1, 2, and 3. The first panel of coefficients can be interpreted as those from a binary logit regression where your dependent variable is recoded as 1 vs. 2+3. The second panel of coefficients can be interpreted as those from a binary logit regression where your dependent variable is recoded as 1 vs. 2+3. The second panel of coefficients can be interpreted as those from a binary logit regression where your dependent variable is recoded as 1 vs. 2+3. The second panel of coefficients can be interpreted as those from a binary logit regression where your dependent variable is recoded 1+2 vs. 3. Positive coefficients mean that higher values on the covariates make higher values on the dependent variable more likely.

Interpretation is particularly straightforward for those variables that meet the parallel lines assumption. From the above we can see that whites and older people tend to be less supportive of working mothers, while those who are better educated and have greater occupation prestige tend to be more supportive. The coefficients for yr89 are consistently positive but decline across cut-points. This means that respondents in 1989 were more supportive of working mothers than respondents in 1977, with the greatest differences being that 1989 respondents were less likely to put themselves in the strongly disagree and disagree categories. Conversely, the male effect is negative but gets larger across cutpoints. Hence, males tend to be less supportive of working mothers than are females, with the greatest differences being that males are less likely to place themselves in the Strongly Agree and Agree categories.

Hence, through the partial proportional odds model estimated by gologit2, the effects of the variables that meet the parallel lines assumption are easily interpretable (you interpret them the same way as you do in ologit). For other variables, an examination of the pattern of coefficients reveals insights that would be obscured or distorted if a proportional odds model were estimated instead. Conversely, an mlogit model might lead to similar conclusions as gologit2 but there would be many more parameters to look at, and the increased number of parameters could cause some effects to become statistically insignificant.

Example 2: Using the gamma option. Here is an example from Lall and colleagues (2002). The dependent variable, hstatus, is measured on a 4 point scale with categories 4 = poor, 3 = fair, 2 = good, 1 = excellent. The independent variables are heart (0 = did not suffer from heart attack, 1 = did suffer from heart attack) and smoke (0 = does not smoke, 1 = does smoke).

Variable			(Good, fair, poor) vs excellent		(Fair, poor) vs (excellent, good)		Poor vs (excellent, good, fair)	
	In(O.R.)	s.e. In(O.R.)	In(O.R.)	s.e. In(O.R.)	In(O.R.)	s.e. In(O.R.)	In(O.R.)	s.e. In(O.R.)
	of log	t component odds ratio ut-off points	Increme	nt at cut-off p	oints			
Suffered from a heart attack (yes/no)?	1.023	0.0554	_	_	—	_	_	_
Do you smoke (yes/no)?	0.1218	0.059	0		0.00822	(0.0628)	0.3382	(0.1006)
			Log odds ratios at cut-off points					
Do you smoke (yes/no)?	—	_	0.1218	(0.059)	0.1300	(0.0991)	0.4600	(0.1281)

Table 5 Log odds ratios for unconstrained partial proportional odds model

In the parameterization of the partial proportional odds model used in their paper, each X has a beta coefficient associated with it (called the "constant component" in the above table). In addition, each X can have M -2 Gamma coefficients (labeled above as the "Increment at cut-off points"), where M = the # of categories for Y and the Gammas represent deviations from proportionality. If the Gammas for a variable are all 0, the variable meets the proportional odds assumption. In the above, there are gammas for smoke but not heart; this means that heart is constrained to meet the proportional odds assumption but smoking is not. *In effect, then, a test of the parallel lines assumption for a variable is a test of whether its gammas equal zero.*

The parameterization used by Lall can be produced by using gologit2's gamma option (with minor differences probably reflecting differences in the software used). Further, by using the autofit option, we can see whether we come up with the same final model that they do.

Generalized Ordered Logit Estimates	Number of obs	=	12535
	LR chi2(4)	=	373.10
	Prob > chi2	=	0.0000
Log likelihood = -14664.661	Pseudo R2	=	0.0126

(1) [Excellent]heart - [Good]heart = 0
(2) [Good]heart - [Fair]heart = 0

hstatus	Coef.	Std. Err.	z	₽> z	[95% Conf.	Interval]
Excellent						
heart	1.025339	.0551397	18.60	0.000	.9172672	1.133411
smoke	.127191	.0590098	2.16	0.031	.0115339	.2428482
_cons	1.303032	.0251244	51.86	0.000	1.253789	1.352275
Good						
heart	1.025339	.0551397	18.60	0.000	.9172672	1.133411
smoke	.1283844	.0488556	2.63	0.009	.0326292	.2241396
_cons	8967713	.0226262	-39.63	0.000	9411177	8524248
Fair						
heart	1.025339	.0551397	18.60	0.000	.9172672	1.133411
smoke	.4581369	.0894379	5.12	0.000	.2828418	.633432
_cons	-3.082652	.0463864	-66.46	0.000	-3.173568	-2.991737

Alternative parameterization: Gammas are deviations from proportionality

hstatus	Coef.	Std. Err.	Z	P> z	[95% Conf.	Interval]
Beta heart	1.025339	.0551397	18.60	0.000	.9172672	1.133411
smoke	.127191 +	.0590098	2.16	0.031	.0115339	.2428482
Gamma_2 smoke	.0011933	.0629692	0.02	0.985	1222239	.1246106
Gamma_3 smoke	.3309459	.100827	3.28	0.001	.1333287	.5285631
Alpha					1 050500	
_cons_1 _cons_2 _cons_3	1.303032 8967713 -3.082652	.0251244 .0226262 .0463864	51.86 -39.63 -66.46	0.000 0.000 0.000	1.253789 9411177 -3.173568	1.352275 8524248 -2.991737

The relationship between these two parameterizations is fairly straightforward. The coefficients for the first equation in the default parameterization correspond to the betas in the alternate parameterization. Gamma_2 parameters = Equation 2 – Equation 1 parameters and Gamma_3 parameters = Equation 3 – Equation 1 parameters. E.g. in equation 3 the coefficient for smoke is .458, and in equation 1 it is .127. Gamma_3 for smoke therefore equals .458 - .127 = .331. You only get gammas for variables that are NOT constrained to meet the proportional odds assumption.

The use of the autofit parameter confirms that Lall got it right, i.e. autofit produces the same partial proportional odds model that he got. But, if we wanted to just trust him, we could have estimated the same model by using the pl or npl parameters. The following two commands will each produce the same results in this case:

. gologit2 hstatus heart smoke, pl(heart) gamma lrf . gologit2 hstatus heart smoke, npl(smoke) gamma lrf Using either parameterization, the results suggest that those who have had heart attacks tend to report worse health. The same is true for smokers, but smokers are especially likely to report themselves as being in poor health as opposed to fair, good or excellent health.

A researcher might want to use the alternative gamma parameterization simply because it is standard practice in their field. But, even if you don't want to report things that way, there are several advantages to at least looking at it.

- You can see at a glance which variables are constrained to have proportional effects and which ones aren't. If there isn't a Gamma parameter, the variable is constrained to meet the proportional odds assumption.
- The printout is more parsimonious. The default parameterization will report the same values multiple times whenever a variable's effect has been constrained to be proportional. The alternate format only reports each parameter once. Note that the Model D.F. corresponds to the number of Betas and Gammas that are reported (unless additional constraints have been applied.)
- By starting with an unconstrained model, the alternate parameterization helps you to see at a glance where the potential problems in a model are. If the gammas for a variable are all statistically insignificant, it is probably safe to impose the proportionality constraint; but if one or more gammas are significant then you probably don't want to impose constraints. This could also lead to models that are even more parsimonious than those estimated by autofit. For example, with the Lall data,

. gologit2 hstatus heart smoke, lrf npl gamma

Generalized Ordered Logit Estimates	Number of obs	=	12535
	LR chi2(6)	=	373.70
	Prob > chi2	=	0.0000
Log likelihood = -14664.362	Pseudo R2	=	0.0126

[default parameterization delete]

Alternative parameterization: Gammas are deviations from proportionality

hstatus	Coef.	Std. Err.	Z	P> z	[95% Conf.	Interval]
Beta	 					
heart	1.046722	.1023646	10.23	0.000	.8460913	1.247353
smoke	.1274032	.0590163	2.16	0.031	.0117334	.2430729
Gamma_2						
heart	0109007	.100116	-0.11	0.913	2071244	.185323
smoke	.0012914	.0629834	0.02	0.984	1221537	.1247365
Gamma_3						
heart	0821184	.1328688	-0.62	0.537	3425365	.1782996
smoke	.3305576	.1007839	3.28	0.001	.1330249	.5280903
Alpha						
cons_1	1.302031	.0254276	51.21	0.000	1.252194	1.351868
_cons_2	8973008	.0228198	-39.32	0.000	9420269	8525748
_cons_3	-3.069089	.0494071	-62.12	0.000	-3.165925	-2.972252

We see that only gamma_3 for smoke significantly differs from 0. Ergo, we could use the constraints option to come up with an even more parsimonious model:

. constraint 1 [#1=#2]:smoke

. gologit2 hstatus heart smoke, lrf gamma pl(heart) constraint(1)

Generalized Ordered Logit Estimates				LR ch	r of obs = i2(3) = > chi2 =	12535 373.10 0.0000
Log likelihood	d = -14664.66		Pseud		0.0126	
(2) [Excel]	lent]smoke - lent]heart - heart - [Fair	[Good]heart				
hstatus	Coef.	Std. Err.	Z	P> z	[95% Conf.	Interval]
Excellent heart smoke cons Good	1.025334 .1279526 1.3029	.055139 .0432192 .024137	18.60 2.96 53.98	0.000 0.003 0.000	.9172638 .0432446 1.255592	1.133405 .2126606 1.350208
heart smoke _cons	1.025334 .1279526 8966838	.055139 .0432192 .0221497	18.60 2.96 -40.48	0.000 0.003 0.000	.9172638 .0432446 9400964	1.133405 .2126606 8532712
Fair heart smoke _cons	1.025334 .4578386 -3.082591	.055139 .0880417 .046273	18.60 5.20 -66.62	0.000 0.000 0.000	.9172638 .28528 -3.173284	1.133405 .6303971 -2.991898
Alternative pa	arameterizati	on: Gammas a	re deviat	ions fro	m proportiona	lity
hstatus	Coef.	Std. Err.	Z	P> z	[95% Conf.	Interval]
Beta heart smoke	1.025334 .1279526	.055139 .0432192	18.60 2.96	0.000 0.003	.9172638 .0432446	1.133405 .2126606
Gamma_2 smoke	 -3.05e-16	6.59e-10	-0.00	1.000	-1.29e-09	1.29e-09
Gamma_3 smoke	.329886	.0838936	3.93	0.000	.1654577	.4943144
Alpha _cons_1 _cons_2 _cons_3	1.3029 8966838 -3.082591	.024137 .0221497 .046273	53.98 -40.48 -66.62	0.000 0.000 0.000	1.255592 9400964 -3.173284	1.350208 8532712 -2.991898

Note that gologit2 is not smart enough to know that Gamma_2 should not be in there (it knows to omit it when either pl or npl have forced the parameter to be 0, but not when the constraint option has been used) but this is just a matter of aesthetics; everything is being done correctly. The fit for this model is virtual identical to the fit of the model that included

gamma 2 (LR chi2 = 373.10 in both), so we conclude that this more parsimonious parameterizations is justified. Hence, while the assumptions of the 2-parameter proportional odds model estimated by ologit are violated by these data, we can get a model that fits whose assumptions are not violated simply by allowing one gamma parameter to differ from 0.

Example 3: svy estimation. The Stata 8 survey data manual presents an example where svyologit is used for an analysis of the NHANES II dataset. The variable health contains self-reported health status, where 1 = poor, 2 = fair, 3 = average, 4 = good, and 5 = excellent. gologit2 can analyze survey data by including the svy parameter. Data must be svyset first. The original example includes variables for age and age^2. To make the results a little more interpretable, I have created centered age (c age) and centered age^2 (c age2) (you need to install Ben Jann's center command to do things the same way I did). This does not change the model selected or the model fit. Note that the lrforce option has no effect when doing svy estimation since likelihood ratio chi-squares are not appropriate in such cases.

. use http://www.stata-press.com/data/r8/nhanes2f.dta . center age . gen c_age2=c_age^2 . gologit2 health female black c_age c_age2, svy auto _____ Testing parallel lines assumption using the .05 level of significance... Step 1: black meets the pl assumption (P Value = 0.2310) Step 2: The following variables do not meet the pl assumption: female (P Value = 0.00280) $c_age (P Value = 0.00000)$ c_age2 (P Value = 0.00004) If you re-estimate this exact same model with gologit2, instead of autofit you can save time by using the parameter pl(black) _____ Generalized Ordered Logit Estimates pweight: finalwgt Strata: stratid PSU: psuid Number of obs=10335Number of strata=31Number of PSUs=62 Population size = 1.170e+08 F(13, 19) = 52.24Prob > F = 0.0000 (1) [poor]black - [fair]black = 0
(2) [fair]black - [average]black = 0

```
( 3) [average]black - [good]black = 0
```

	health	Coef.	Std. Err.	t	P> t	[95% Conf.	Interval]
poor		+					
-	female	.1681817	.1034177	1.63	0.114	0427401	.3791034
	black	-1.008808	.0836513	-12.06	0.000	-1.179416	8382006
	c_age	0617038	.003537	-17.45	0.000	0689175	05449
	c_age2	.0006893	.0003049	2.26	0.031	.0000674	.0013111
	_cons	2.962162	.1373065	21.57	0.000	2.682124	3.242201
fair		+					
	female	1545385	.0680284	-2.27	0.030	2932834	0157937
	black	-1.008808	.0836513	-12.06	0.000	-1.179416	8382006
	c_age	0525504	.002082	-25.24	0.000	0567966	0483042
	c_age2	000028	.0001237	-0.23	0.822	0002802	.0002242
	_cons	1.718909	.0765319	22.46	0.000	1.562821	1.874997
avera	ige	+					
	female	1576817	.0596012	-2.65	0.013	279239	0361243
	black	-1.008808	.0836513	-12.06	0.000	-1.179416	8382006
	c_age	0409575	.0017576	-23.30	0.000	0445422	0373728
	c_age2	8.91e-06	.0000882	0.10	0.920	000171	.0001889
	_cons	.1705633	.0534477	3.19	0.003	.0615559	.2795707
good		+					
	female	2133394	.0636419	-3.35	0.002	3431379	0835408
	black	-1.008808	.0836513	-12.06	0.000	-1.179416	8382006
	c_age	0356466	.0020002	-17.82	0.000	039726	0315672
	c_age2	0004546	.0001311	-3.47	0.002	0007221	0001872
	_cons	9136691	.0574078	-15.92	0.000	-1.030753	7965851

In this example, only one variable, black, meets the parallel lines assumption. Blacks tend to report worse health than do whites. For females, the pattern is more complicated. They are less likely to report poor health than are males (see the positive female coefficient in the poor panel), but they are also less likely to report higher levels of health (see the negative female coefficients in the other panels), i.e. women tend to be less at the extremes of health than men are. Such a pattern would be obscured in a straight proportional odds model. The effect of age is more extreme on lower levels of health.

Example 4. gologit 1.0 compatibility. Some post-estimation commands – specifically, the spost routines of Long and Freese – currently work with the original gologit but not gologit2. That should change in the future. For now, you can use the v1 parameter to make the stored results from gologit2 compatible with gologit 1.0. (Note, however, that this may make the results non-compatible with post-estimation routines written for gologit2, including predict.) Using the working mother's data again,

```
. use http://www.nd.edu/~rwilliam/stata/ordwarm2, clear
(77 & 89 General Social Survey)
```

```
. * Use the v1 option to save internally stored results in gologit 1.0 format . quietly gologit2 warm yr89 male white age ed prst, p1(yr89 male) lrf v1
```

. * Use spost routines. Get predicted probability for a 30 year old average white woman in 1989 . prvalue, x(male=0 yr89=1 age=30) rest(mean) gologit: Predictions for warm Predicted probabilities for each category: Pr(y=SD|x): 0.0473 0.1699 0.4487 0.3340 Pr(y=D|x):Pr(y=A|x):Pr(y=SA|x): yr89malewhiteageed10.87658093012.21805530 prst 30 12.218055 39.585259 x= . * Now do 70 year old average black male in 1977 . prvalue, x(male=1 yr89=0 age=70) rest(mean) gologit: Predictions for warm Predicted probabilities for each category: Pr(y=SD|x): 0.2565 $\Pr(y=D|x):$ 0.4699 0.2093 0.0644 Pr(y=A|x):Pr(y=SA|x): yr89 male 0 1 malewhiteageedprst1.87658097012.21805539.585259 x=

These "representative" cases show us that a 30 year old average white woman in 1989 was much more supportive of working mothers than a 70 year old average black male in 1977. Various other spost routines that work with the original gologit (not all do) can also be used, e.g. prtab.

Example 5: The predict command. In addition to the standard options (xb, stdp, stddp) the predict command supports the pr option (abbreviated p) for predicted probabilities; pr is the default option if nothing else is specified. For example,

Author

Richard Williams Notre Dame Department of Sociology <u>Richard.A.Williams.5@ND.Edu</u> <u>http://www.nd.edu/~rwilliam</u>

Acknowledgements

Vincent Kang Fu of the Utah Department of Sociology wrote gologit 1.0 and graciously allowed Richard Williams to incorporate parts of its source code and documentation in gologit2.

The documentation for Stata 8.2's mlogit command and the program mlogit_p were major aids in developing the gologit2 documentation and in adding support for the predict command. Much of the code is adapted from *Maximum Likelihood Estimation with Stata*, Second Edition, by William Gould, Jeffrey Pitblado and William Sribney.

Sarah Mustillo, Dan Powers, J. Scott Long, and Nick Cox provided stimulating and helpful comments.

References

Fu, Vincent. 1998. "Estimating Generalized Ordered Logit Models." *Stata Technical Bulletin* 8:160-164.

Lall, R., S.J. Walters, K. Morgan, and MRC CFAS Co-operative Institute of Public Health. 2002. "A Review of Ordinal Regression Models Applied on Health-Related Quality of Life Assessments." *Statistical Methods in Medical Research* 11:49-67.

Long, J. Scott and Jeremy Freese. 2003. *Regression Models for Categorical Dependent Variables Using Stata, 2nd Edition.*

Peterson, Bercedis and Frank E. Harrell Jr. 1990. "Partial Proportional Odds Models for Ordinal Response Variables." *Applied Statistics* 39(2):205-217.

Suggested citation if using gologit2 in published work (at least until something more formal comes along):

Williams, Richard. 2005. "Gologit2: A Program for Generalized Logistic Regression/ Partial Proportional Odds Models for Ordinal Variables." Retrieved May 12, 2005 (http://www.nd.edu/~rwilliam/stata/gologit2.pdf).